
Installation
    

To create an ECSx application, there are a few simple steps:
	install Elixir + erlang/OTP
	install Phoenix (optional)
	create an Elixir/Phoenix project
	fetch ECSx as a dependency for your project
	run the ECSx setup

Elixir and erlang/OTP
If you don't yet have Elixir and erlang/OTP installed on your machine, follow the instructions on the official Installation Page.  
Phoenix
If you plan on hosting your application online, you'll probably want to use Phoenix.  You can skip this step if you only want to run the app locally.  Otherwise, follow the instructions for Phoenix installation (the tutorial project will assume you are using Phoenix).
Create project
If you are using Phoenix, you'll create your new application with the command
$ mix phx.new my_app

Or for a regular Elixir application (with supervision tree):
$ mix new my_app --sup

Install ECSx
To use the ECSx framework in your application, it should be added to the list of dependencies in my_app/mix.exs:
defp deps do
  [
    {:ecsx, "~> 0.1"}
  ]
end
Then (from the root directory of your application) run:
$ mix deps.get

Setup ECSx
With ECSx installed, you can run the setup generator:
$ mix ecsx.setup

which will create the Manager, and two folders to get your project started.
You'll also need to add the Manager to your application's supervision tree:
def start(_type, _args) do
  children = [
    MyApp.Manager
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end
Summary
You should now have everything you need to start building!  If you're already familiar with the Entity-Component-System pattern, jump right in to the tutorial project - otherwise, start with our guide on ECS design.



  

    
Tutorial Project
    

Now that we have an Elixir project with Phoenix, we can get started on building a game with
Entity-Component-System architecture.  We're going to use the classic Snake as inspiration.
Design plan
In a classic Snake game, collision is everything.  We're going to need a game world where a snake
can occupy many coordinates, and each one of those coordinates should be considered for collision.
We also need to "order" each occupied coordinate in order to simulate movement - by adding to the
head and removing from the tail.
Let's start by creating a Coordinate aspect:
  $ mix ecsx.gen.aspect Coordinate coordinate snake_id order

We are declaring three fields in the schema, coordinate, snake_id, and order.  Normally,
the first field of a schema is entity_id, which will give us efficient lookups on that field.
However, this is a rare exception where lookup by entity_id is not very useful, and it is
coordinate which will be the basis for our querying.  The id of the snake is still useful
metadata, so we'll include that as the second field.  The third field is unique to our Snake game,
where each coordinate represents just one link to a longer chain that is a snake.  We need to be
able to know the first (head) and last (tail) coordinates, and we accomplish this through order.
A coordinate with order 1 is the head, and a coordinate where order is equal to the total
length of the snake, must be the tail.
Since we know that snake length will be useful data, let's create another Aspect for it:
  $ mix ecsx.gen.aspect Length snake_id length

This is a standard Aspect where we will query its Components by snake_id, so that should be
the first field.
Next we can also anticipate needing to label snakes with a Direction of movement:
  $ mix ecsx.gen.aspect Direction snake_id direction

Now that we have modeled world data and snake data, let's think about the Systems which will
organize game logic.  What makes a Snake game work?
	Snakes move forwards every game tick
	Snakes get longer over time (or based on other game conditions)
	When there is a collision, one or both snakes are removed from the game

Let's start with the most important System - the physics.  We'll call it Driver:
  $ mix ecsx.gen.system Driver

Heading over to the generated file lib/your_app/systems/driver.ex and we'll add some code:
defmodule YourApp.Systems.Driver do
  ...
  alias YourApp.Aspects.Moving
  alias YourApp.Aspects.Position
  ...
  def run do
    # First we get all the Coordinate components
    coordinates = Coordinate.query_all()

    # Each coordinate will be appropriately updated to simulate the movement of the snakes
    Enum.each(coordinates, &update_coordinate/1)
  end

  defp update_coordinate(%{coordinate: {x, y}, snake_id: id, order: 1} = component) do
    # If the order is 1, then this is a head;  we need to occupy an adjacent coordinate.
    # First let's find the direction this snake is headed
    direction = Direction.query_one(match: [snake_id: id], value: :direction)

    # From this direction we need to calculate the coordinate we're moving into
    new_coord = calculate_new_position(x, y, direction)

    # Insert new head coordinate and increment the order of this one
    Coordinate.add_component(coordinate: new_coord, snake_id: id, order: 1)
    increment_order(component)
  end

  defp update_coordinate(%{coordinate: coordinate, snake_id: id, order: order} = component) do
    # Since we know this isn't the head, we just need to check if it's the tail
    length = Length.query_one(match: [snake_id: id], value: :length)

    if length == order do
      # This is the tail, we should un-occupy the coordinate as the snake moves out
      Coordinate.remove_component(coordinate: coordinate, snake_id: id)
    else
      # All other coordinates in-between get their order incremented by one
      increment_order(component)
    end
  end

  defp calculate_new_position(x, y, :north), do: {x, y + 1}
  defp calculate_new_position(x, y, :east), do: {x + 1, y}
  defp calculate_new_position(x, y, :south), do: {x, y - 1}
  defp calculate_new_position(x, y, :west), do: {x - 1, y}

  defp increment_order(%{coordinate: coordinate, snake_id: id, order: order}) do
    Coordinate.remove_component(coordinate: coordinate, snake_id: id)
    Coordinate.add_component(coordinate: coordinate, snake_id: id, order: order + 1)
  end
end
You probably noticed that this System creates new coordinates without checking if there is any
collision with existing coordinates.  This is intentional;  to demonstrate why, imagine an
example where Snake A is exiting a coordinate, and Snake B is entering the same coordinate,
on the same server tick.  Now, if we check for collision in the Driver system, the result will
depend on which Component gets updated first:
	If Snake A's tail Component is updated first, then the check will show the coordinate as
unoccupied, and there will be no collision.
	If Snake B's head Component is updated first, then the check will show the coordinate as
occupied by Snake A's tail, and there will be a collision.

We want to avoid this kind of inconsistency, and ensure that the result is the same, regardless
of which Component is stored earlier in the table.  Therefore we allow duplicate coordinates,
and will have another System handle the cleanup afterwards.
You've probably guessed that we'll start by running the generator:
  $ mix ecsx.gen.system Collision

But before we start coding, let's think of a plan for how to efficiently check for collisions.
One approach could be to fetch all the Coordinate Components, iterate over the list, grouping
them by {x, y} pair, then iterate over the groups, checking if any have more than one member.
This might be fine for some games, but if we want to optimize performance, we should only check
for collision where there is actually a possibility of collision.  The only possible points of
collision are those coordinates where there is a snake head (order 1).
Then, in lib/your_app/systems/collision.ex:
defmodule YourApp.Systems.Collision do
  ...
  def run do
    # Fetch coordinates for all snake heads
    possible_collision_coords = Coordinate.query_all(match: [order: 1])
    # Update components for any entities which have collided
    Enum.each(possible_collision_coords, &check_for_collision/1)
  end

  defp check_for_collision(%{coordinate: coordinate, snake_id: id, order: order}) do
    case Coordinate.query_all(match: [coordinate: coordinate]) do
      [] -> :ok
      [_] -> :ok
      multiple_results -> Enum.each(multiple_results, &handle_collision/1)
    end
  end

  defp handle_collision(%{coordinate: coordinate, snake_id: id, order: 1}) do
    # When a snake head collides, it dies - we can remove its components
    Position.remove_component(id)
    Moving.remove_component(id)
    Length.remove_component(id)

    # Without any components, the entity will cease to exist!
    # Maybe we would like to keep some record of the entity instead:
    CrashRecord.add_component(
      entity_id: id,
      crash_time: DateTime.utc_now(),
      crash_location: coordinate
    )
  end

  # If the order is not 1, then the collision was on the snake's tail, and it will survive
  defp handle_collision(_), do: :ok
end
Whenever we need a new Aspect (such as CrashRecord), we can simply run the generator again:
  $ mix ecsx.gen.aspect CrashRecord entity_id crash_time crash_location




  

    
ECS Design
    

Entities, Aspects, and Components
Everything in your application is an Entity, but in ECS you won't work with these
Entities directly - instead you will work directly with the individual attributes
that an Entity might have.  These attributes are called Aspects, and they are given
to an Entity by creating a Component, which holds, at minimum, the Entity's
unique ID, but also any extra data which is relevant to that Aspect.  For example:
	You're running a simulation of cars on a highway
	Each car gets its own entity_id e.g. 123
	If the car with ID 123 is blue, we give it the Color Aspect, stored as a {123, "blue"} Component
	If the same car is moving west at 60mph, we might model this with a Moving Aspect, stored as a {123, 60, "west"} Component
	Another car with ID 135 might have the Moving Aspect with different data {135, 30, "east"}
	These cars would also have Location which holds x, y, z coordinates: {entity_id, x, y, z}

Systems
Once your data is modeled using Components to associate different Aspects to your Entities,
you'll create Systems to operate on them.  For example:
	Entities with the Moving Aspect should have their locations regularly updated according to the speed and direction
	We can create a Move System which reads the Moving Component, calculates how far the car has moved since the last server tick, and updates the Entity's Location Component accordingly.
	The System will run every tick, only considering Entities which have the Moving Aspect




  

    
ECSx 
    



      
ECSx is an Entity-Component-System (ECS) framework for Elixir.
In ECS:
	Every game object is an Entity, represented by a unique ID.
	The data which comprises an Entity is split among many Components.
	Game logic is split into Systems, which update the Components every server tick.

Components are grouped together - with similar Components from other Entities - into
categories called Aspects.  Systems can then be organized in a way where each one only
operates on Components of a specific Aspect.
Under the hood, ECSx uses Erlang Term Storage (ETS) to store active Components in memory.
A single GenServer manages the ETS tables to ensure strict serializability and customize
the run order for Systems.

      





  

    
ECSx.Aspect behaviour
    



      
Provides an API for working with Components of a specific type.
For example, if Entities in your application should have a "color" value, you will
create an Aspect called Color.  This allows you to add a color Component to an Entity
with add_component/1, query the color value for a given Entity with query_one/1,
query all Entities which have a color value with query_all/1, remove
the color value from an Entity altogether with remove_component/1, or test whether
an entity has a color with has_component?/1.
Usage
Each Aspect should have its own module, which defines a schema for the data, and optional
ETS table settings.  The schema defines the field names which will be used to reference
each value.
defmodule MyApp.Aspects.Color do
  use ECSx.Aspect,
    schema: {:entity_id, :hue, :saturation, :lightness}
end
Options
	:schema - a tuple of field names
	:table_type - ETS table type.  Possible options are :set, :ordered_set, :bag, and
:duplicate_bag.  Defaults to :set.  See below for more information.
	:read_concurrency - when true, enables read concurrency for the ETS table.  Defaults
to false

:table_type option
By default, ECSx creates ETS tables of type :set for your Components.  This means that
each Entity may have at most one Component of each Aspect.  For example, if your application
has a "Height" Aspect, Entities would never need more than one Component to model that value.
However, if you had an Aspect such as "TakingDamageOverTime", you might want an Entity to
store each source of damage as a separate value, using multiple Components of the same Aspect.
To do this, set :table_type to :bag.
Example
defmodule TakingDamageOverTime do
  use ECSx.Aspect,
    schema: {:entity_id, :damage_per_second, :damage_type, :source_id},
    table_type: :bag
end

alias TakingDamageOverTime, as: DOT

DOT.add_component(entity_id: hero.id, damage_per_second: 10, type: :poison, source_id: spider.id)
DOT.add_component(entity_id: hero.id, damage_per_second: 25, type: :fire, source_id: dragon.id)
query_one/1 and query_all/1
The standard way to fetch the Component(s) from an Entity is using a Query.  Each aspect
provides two Query functions: query_one/1 and query_all/1.  The former returns a single
result, and will raise an ECSx.QueryError if more than one result is found.  The latter
will return a list with any number of results.
These Query functions have two possible parameters:
	:match - A keyword list of the fields and values for which to search.  If a :match is
not given, the entire Aspect table will be returned.

	:value - By default, each Component returned by a Query will be in the form of a map,
using the keys provided by its Aspect schema.  If you only care about one field from the
Component, you can instead return unwrapped values with value: :field_name


Examples
Color.query_one(match: [entity_id: entity_id])
%{entity_id: entity_id, hue: 300, saturation: 50, lightness: 45}

Color.query_one(match: [entity_id: entity_id], value: :hue)
300

Color.query_all()
[%{entity_id: entity_id, ...}, %{...}, ...]

Color.query_all(value: :entity_id)
[entity_id, another_entity_id, ...]

          Creates a new component.



      


      
        
          has_component?(entity_id)

        


          Checks if an entity has one or more components with this aspect.



      


      
        
          query_all(query)

        


          Query for all components of this aspect with optional match conditions.



      


      
        
          query_one(query)

        


          Query for a single component of this aspect with optional match conditions.



      


      
        
          remove_component(entity_id)

        


          Removes any existing components of this aspect from an entity.

          @callback add_component(attrs :: Keyword.t()) :: :ok


      


Creates a new component.

  
  example

  
  Example


ArmorRating.add_component(entity_id: 123, value: 10)

  



  
    
      
      Link to this callback
    
    has_component?(entity_id)

          @callback has_component?(entity_id :: any()) :: boolean()


      


Checks if an entity has one or more components with this aspect.

  



  
    
      
      Link to this callback
    
    query_all(query)

          @callback query_all(query :: Keyword.t()) :: [ECSx.Component.t() | ECSx.Component.value()]


      


Query for all components of this aspect with optional match conditions.
Examples
# Get the ID of each entity with zero velocity in the x-y plane
Velocity.query_all(match: [vx: 0, vy: 0], value: :entity_id)

  



  
    
      
      Link to this callback
    
    query_one(query)

          @callback query_one(query :: Keyword.t()) :: ECSx.Component.t() | ECSx.Component.value()


      


Query for a single component of this aspect with optional match conditions.
Raises if more than one component is returned.
Examples
# Get the Velocity component for entity 123
Velocity.query_one(match: [entity_id: 123])

  



  
    
      
      Link to this callback
    
    remove_component(entity_id)

          @callback remove_component(entity_id :: any()) :: :ok


      


Removes any existing components of this aspect from an entity.

  


        

      



  

    
ECSx.Component 
    



      
A Component labels an entity as possessing a particular Aspect, and holds any data
needed to model that Aspect.
Under the hood, we use ETS to store the Components in memory for quick retrieval
via Aspect and Entity ID.

ECSx.Manager 
    



      
The Manager for your ECSx application.
In an ECSx application, the Manager is responsible for:
	starting up ETS tables for each Aspect, where the Components will be stored
	prepopulating the game content into memory
	keeping track of the Systems to run, and their run order
	configuring the tick rate for the application
	running the Systems every tick

:tick_rate option
The mix ecsx.setup generator creates a Manager file with :tick_rate set to 20
(i.e. 20 ticks per second).  Feel free to change this number to fit the needs
of your application.
aspects/0 and systems/0
Your Manager module must contain two zero-arity functions called aspects and systems
which return a list of all Aspects or Systems in your application.  The order of
the Aspects list is irrelevant, but the order of the Systems list is very important,
because the Systems are run consecutively in the given order.
setup block
Another important piece of the Manager module is the setup block.  Here you can load
all the necessary data for your app before any Systems run or users connect.  See setup/1
for more information.

      Link to this macro
    
    setup(block)


      
       
       View Source
     


      (macro)

  


  

Runs the given code block during startup.
The code will be run during the Manager's initialization (so pay special attention to the
position of ECSx.Manager in your application's supervision tree). The Aspect tables will
be created before setup is executed.

  
  example

  
  Example


defmodule YourApp.Manager do
  use ECSx.Manager

  setup do
    for npc <- YourApp.fetch_npc_spawn_info() do
      Name.add_component(entity_id: npc.id, name: npc.name)
      HitPoints.add_component(entity_id: npc.id, current: npc.hp, max: npc.hp)
      Location.add_component(entity_id: npc.id, coords: npc.spawn_location)
    end
  end
end
This setup will spawn each NPC with Components for Name, HitPoints, and Location.

  


        

      



  

    
ECSx.System behaviour
    



      
A fragment of game logic which reads and updates Components.
Every System must implement a run function.
By default, the System will run every game tick.  To use a longer period between runs,
you can pass the option :period.  For example, to set a System to run every 5 ticks:
use ECSx.System,
  period: 5

          run()

        


          Invoked to run System logic.

      Link to this callback
    
    run()

          @callback run() :: :ok


      


Invoked to run System logic.
This function will be called every T game ticks, where T is the value of
the System's :period option (defaults to 1).
Note:  A crash inside this function will restart the entire app!

  


        

      



  

    
ECSx.QueryError exception
    



      
Raised at runtime when the Query is invalid.

      





  

    
mix ecsx.gen.aspect 
    



      
Generates a new Aspect for an ECSx application.
$ mix ecsx.gen.aspect Color entity_id hue saturation lightness

The first argument is the name of the Aspect module, followed by the data fields
which will make up each Component.

      





  

    
mix ecsx.gen.system 
    



      
Generates a new System for an ECSx application.
$ mix ecsx.gen.system Foo

The only argument accepted is a module name for the System.

      





  

    
mix ecsx.setup 
    



      
Generates the Manager process which runs an ECSx application.
$ mix ecsx.setup

This setup will generate manager.ex and empty folders for aspects and systems.
If you don't want to generate the folders, you can provide option --no-folders

